

LANGUAGE	MODULE	REVISION	DRAWING NO.
العربية	أساسيات المتحكمات الدقيقة	A	BP-001

بنية مصفوفات الاستشعار المستقلة

توصيل وتغذية وبرمجة نظام متعدد المستشعرات على متحكم دقيق واحد – المهارة الأساسية التي يقوم عليها معظم مشاريع معارض العلوم القائمة على الأجهزة.

التكلفة التقريبية: \$18-28

مدة التنفيذ: ٢-٣ ساعات

المستوى: مبتدئ إلى متوسط

01 الهدف

بنهاية هذا المخطط، سيكون لديك متحكم Arduino Uno واحد يقرأ في آنٍ واحد من ثلاثة أنواع مختلفة من المستشعرات – المسافة، الحرارة والرطوبة، والضوء – ويطلع القراءات بشكل منظم وواضح على شاشة الحاسوب. والأهم من ذلك، ستفهم لماذا تم تنظيم الكود بهذا الشكل: كل مستشعر له دالة قراءة خاصة به، بحيث يمكن للنظام أن يتوسع من ٣ مستشعرات إلى ٣٠ دون أن يصبح الكود معقدًا أو صعب الإدارة. هذا النمط – دالة واحدة لكل مستشعر – هو "البنية" المقصودة في عنوان هذا المخطط، وهو نفس النمط المستخدم في أنظمة المراقبة البيئية والروبوتات الحقيقية.

02 المتطلبات الأساسية

- تثبيت برنامج Arduino IDE على حاسوبك (مجاني، من الموقع الرسمي [arduino.cc](https://www.arduino.cc))
- فهم أساسي للفرق بين المنافذ الرقمية (تشغيل/إيقاف) والمنافذ التناظرية (نطاق من القيم) – إذا كان هذا مفهوماً جديداً عليك، خصص ١٥ دقيقة لتجربة الأمثلة الجاهزة Blink و AnalogReadSerial قبل البدء
- كابل USB مناسب للوحتك (تأكد أنه كابل بيانات وليس كابل شحن فقط قبل البدء)

03 قائمة المكونات

المكوّن	العدد	التكلفة التقريبية	ملاحظات
Arduino Uno R3 (أو نسخة متوافقة)	١	\$6-10	النسخ البديلة تعمل بنفس الكفاءة ومتوفرة بسهولة في متاجر الإلكترونيات الإقليمية
لوحة تجارب (breadboard) نصف الحجم	١	\$2-3	
مستشعر مسافة بالموجات فوق الصوتية HC-SR04	١	\$1-2	
مستشعر حرارة ورطوبة DHT11	١	\$2-3	يعمل DHT22 أيضًا بتعديل سطر واحد في الكود؛ DHT11 أرخص وكافٍ لهذا المشروع

		مقاومة ضوئية (LDR)	1	<\$1
تُستخدم لتثبيت قراءة المقاومة الضوئية		مقاومة 10kΩ	1	<\$1
		أسلاك توصيل (ذكر-ذكر)	10~	\$2-3
تأكد أنه كابل بيانات وليس كابل شحن فقط		كابل USB-A إلى USB-B	1	\$2-4

04 الأدوات المطلوبة


- حاسوب مثبت عليه برنامج Arduino IDE
- لا حاجة للحام (soldering) في هذا المشروع

ملاحظات السلامة

يعمل هذا المشروع بالكامل على تغذية USB بجهد ٥ فولت — لا يوجد خطر صعق كهربائي حقيقي. مع ذلك، احرص على عادتین مهمتين: (١) افصل كابل USB دائمًا قبل تعديل أي توصيلة، حتى لا يحدث تماس بين منفذين أثناء تشغيل اللوحة، و(٢) تأكد من قطبية التغذية (+ / -) على لوحة التجارب قبل توصيل أي مستشعر، لأن عكس القطبية قد يتلف المستشعر بشكل دائم حتى عند الجهد المنخفض.

05 البنية العامة للنظام

تشارك المستشعرات الثلاثة في نفس خطي التغذية (5V و GND) على لوحة التجارب، ثم يتصل كل مستشعر بمنغذه الخاص كما هو موضح أدناه.



شكل ١ — توصيلات مصفوفة المستشعرات: تخصيص المنافذ وخط التغذية المشترك

التغذية	منفذ (منافذ) الإشارة	المستشعر
5V / GND	Trig → D9, Echo → D10	HC-SR04 (موجات فوق صوتية)
5V / GND	Data → D7	DHT11 (حرارة/رطوبة)
5V / GND	Signal → A0 (10 kΩ مع مقاومة جهد)	LDR (ضوء)

1. ضع الأردوينو ولوحة التجارب جنبًا إلى جنب. صل سلكًا من 5V على الأردوينو إلى الخط الأحمر (+) على لوحة التجارب، ومن GND إلى الخط الأزرق (-). كل مستشعر سيأخذ تغذيته من هذين الخطين.
2. وُصل مستشعر HC-SR04. VCC → الخط الأحمر، GND → الخط الأزرق، Trig → منفذ D9، Echo → منفذ D10.
3. وُصل مستشعر DHT11. VCC → الخط الأحمر، GND → الخط الأزرق، Data → منفذ D7. إذا لم تكن وحدة DHT11 لديك تحتوي على مقاومة سحب مدمجة، أضف مقاومة 10kΩ بين Data و VCC.
4. وُصل LDR كمقسم جهد. صل أحد طرفي LDR بالخط الأحمر (+). صل الطرف الآخر بمنفذ A0 وأيضا بأحد طرفي مقاومة 10kΩ. صل الطرف الآخر للمقاومة بالخط الأزرق (-). هذا المقسم هو ما يحوّل تغيّر مقاومة LDR إلى جهد قابل للقراءة.
5. تحقق من جميع التوصيلات مقارنة بجدول المنافذ في القسم ٥ قبل توصيل التغذية.
6. وُصل USB وارفع الكود من القسم ٧. افتح Serial Monitor (من القائمة Tools → Serial Monitor، بمعدل 9600 baud لرؤية القراءات الحية).

07 شرح الكود

يتبع الكود أدناه نمط "دالة واحدة لكل مستشعر" – لاحظ أن دالة loop() نفسها قصيرة وواضحة، لأن تعقيد كل مستشعر محصور داخل دالته الخاصة. هذا هو النمط الذي يجب الحفاظ عليه عند إضافة مستشعرات جديدة في مشاريع مستقبلية.

```
// MMI Blueprint BP-001 – Autonomous Sensor Array
// Requires: DHT sensor library (install via Library Manager)
#include <DHT.h>

#define TRIG_PIN 9
#define ECHO_PIN 10
#define DHT_PIN 7
#define LDR_PIN A0

DHT dht(DHT_PIN, DHT11);

void setup() {
  Serial.begin(9600);
  pinMode(TRIG_PIN, OUTPUT);
  pinMode(ECHO_PIN, INPUT);
  dht.begin();
}

// -- One function per sensor: the core architecture pattern --
float readDistanceCM() {
  digitalWrite(TRIG_PIN, LOW); delayMicroseconds(2);
  digitalWrite(TRIG_PIN, HIGH); delayMicroseconds(10);
  digitalWrite(TRIG_PIN, LOW);
  long duration = pulseIn(ECHO_PIN, HIGH);
```

```

return duration * 0.0343 / 2;
}

float readLightPercent() {
  int raw = analogRead(LDR_PIN);
  return (raw / 1023.0) * 100.0;
}

void loop() {
  float distance = readDistanceCM();
  float humidity = dht.readHumidity();
  float temp = dht.readTemperature();
  float light = readLightPercent();

  Serial.print("Distance: "); Serial.print(distance); Serial.println(" cm");
  Serial.print("Temp: "); Serial.print(temp); Serial.println(" C");
  Serial.print("Humidity: "); Serial.print(humidity); Serial.println(" %");
  Serial.print("Light: "); Serial.print(light); Serial.println(" %");
  Serial.println("---");

  delay(1000);
}

```

ملاحظة: قد تُرجع الدالتان `dht.readHumidity()` و `dht.readTemperature()` قيمة `nan` أحياناً إذا كان التوقيت غير دقيق — هذا أمر طبيعي مع DHT11 ومُعالج في جدول استكشاف الأخطاء أدناه.

08 الاختبار واستكشاف الأخطاء

العارض	السبب المحتمل	الحل
لا شيء يظهر في Serial Monitor	معدل baud خاطئ، أو منفذ/لوحة غير صحيحة	اضبط Serial Monitor على 9600 baud، وتحقق من Tools → Port
قراءة المسافة دائماً ٠ أو رقم ضخم	تبديل منفذي Trig/Echo، أو سلك غير محكم	راجع التوصيلات مقارنة بجدول القسم ٥.
الحرارة/الرطوبة تظهر "nan"	يحتاج DHT11 فاصلاً ~٢ ثانية بين القراءات	زد قيمة <code>delay()</code> في نهاية <code>loop()</code> إلى ٢٠٠٠ ميلي ثانية على الأقل أثناء الاختبار
قراءة الضوء ثابتة عند ٠ أو ١٠٠٪	مقسم الجهد موصل بشكل معكوس	تأكد أن ترتيب LDR والمقاومة يطابق الخطوة ٤ بالضبط
الحاسوب لا يتعرف على الأردوينو	كابل USB للشحن فقط، أو تعريفات ناقصة	جرب كابلاً مختلفاً أولاً — هذا هو السبب الأكثر شيوعاً

09 الخطوة التالية

من هذا المخطط إلى مشروع بمستوى ISEF

مصفوفة المستشعرات العاملة هي مكوّن، وليست مشروعاً بحد ذاتها بعد. الخطوة التي تحوّلها إلى مشروع بمستوى ISEF هي اختيار مشكلة حقيقية يمكن قياسها — مثل: مراقبة ظروف بيت محمي زراعي لمزرعة محلية، أو تتبع جودة الهواء قرب مدرستك، أو بناء نظام إنذار مبكر لخطر محدد في مجتمعك. اجمع هذا المخطط مع

BP-002 (تسجيل البيانات اللحظي) لبدء تسجيل تلك البيانات عبر الزمن – وهذا ما يحوّل "بنييت مستشعراً" إلى "جمعت أدلة لفرضية علمية"، وهي اللغة التي يكافئها تحكيم ISEF فعلياً. راجع وحدة دليل ISEF والجامعات لمعرفة كيفية صياغة هذا رسمياً كملخص بحثي.